## LICENCE EEA

3 - VHDL

## Systemes Numeriques & ROCESSEURS EMBARQUES



Julien DENOULET

julien.denoulet@sorbonne-universite.fr L3 EEA - LU3EE100


SU - LIP6

#### Plan de Cours

- Intro
- Concepts de Base
- Niveaux de Description VHDL concurrent
- Process VHDL séquentiel
- Comment bien coder en VHDL
- Testbench Simulation
- Annexes: Codes VHDL de modules de base

## Réalisation de Système Num.

- A (not so) very long time ago...
  - Logique câblée
  - Associations de composants discrets
  - Outils de conception:
    - Logiciel de routage de carte






**C**3

## Réalisation de Système Num.

- Some time ago...
  - Logique programmable
  - Apparition de composants programmables de faible ou moyenne densité (PAL, CPLD)



- Outils de conception
  - Saisie de schémas
- Langages de description bas niveau (KARL, ABEL)





**C**3

## Réalisation de Système Num.

#### ■ Now...

- Augmentation exponentielle de la complexité des circuits
- Outils de conception
  - Logiciels de CAO complexes
  - Langages de description de haut niveau (VHDL, VERILOG)





**C**3

## Logiciels

- Xilinx Vivado Design Suite HLx Editions
  - Chaîne de conception et d'implémentation
  - Webpack: Version gratuite (20 Gigas !!)
- Version 2018.3 Utilisé en TP

https://www.xilinx.com/support/download/index.html/content/xilinx/en/downloadNav/vivado-design-tools/2018-3.html

**C**3




## Logiciels

- Autrs solutions
  - Modelsim Student Edition
  - EDA Playground
  - GHDL/GTKWave
  - **...**


Source: Dreamstime.com

 Consulter le document VHDL@Home sur Moodle pour plus d'infos (section TP du site d'UE)



# VHDL Concepts de Base


## Langage de Description





#### "All you can write is what you see..."

Source: Hulton Archive



C3

## Langage HDL

- Son but est... de décrire du matériel et d'aboutir à la réalisation d'un circuit
- Spécificités d'un circuit matériel

Parallélisme







**C3** 




 Permanence des ressources utilisées



#### VHDL is...

- Apparition en 1987
- Mise à jour du standard en 1993, 2002 puis 2008
- But: Synthèse des circuits numériques
  - Description synthétisable
  - Plusieurs niveaux de description
  - RTL: Register Transfer Level



Point d'entrée du flot de conception

**C**3

### VHDL is not...

- Approche différente d'un langage informatique "classique" (C)
  - C: langage séquentiel
    - Une instruction s'exécute après une autre
  - VHDL: langage concurrent
    - Dans un système matériel, tous les blocs travaillent en parallèle
    - Les instructions d'un langage VHDL s'exécutent en parallèle

**C**3


## Description synthétisable

- Un code VHDL peut compiler correctement
  - Car il est syntaxiquement juste
  - Mais ne pas fonctionner en simulation


(Comme en C...)

- Un code VHDL peut correctement fonctionner en simulation
  - Mais être rejeté par un outil de synthèse
  - Car la description n'est pas synthétisable


#### Blocs de base VHDL



#### Blocs de Base VHDL



#### Blocs de Base VHDL



## Exemple

- Déclaration d'une Bibliothèque
  - En tête de programme
  - Utilisation de certains packages
    - Un package regroupe des fonctions ou types de données

```
library ieee;

use ieee.std_logic_1164.all;

use ieee.std_logic_unsigned.all;


Utilisation de deux

packages de la
librairie ieee

Utilisation de toutes les fonctions
du package
```


**C**3

#### Entité



C3

#### Architecture



#### **Ports**

- Le port permet de rentrer ou de sortir une donnée dans le module que l'on décrit
- Le type de la donnée est précisé lors de la déclaration du port

**C3** 

3 types de ports


```
Entrée : TOTO: in std_logic;
```

- Sortie : TITI: out std\_logic;
- Bidirectionnel : TUTU: inout std\_logic;
  - A n'utiliser que si on veut VRAIMENT faire du bidrectionnel (description de bus, etc...)

## Affectation de ports

#### • QU'EST-CE QU'ON NE PEUT PAS FAIRE?

- On ne peut pas affecter (écrire) une valeur sur un port d'entrée
- On ne peut pas lire la valeur d'un port de sortie pour l'affecter à un signal ou un autre port



**C**3

## Signaux

- Un signal peut être vu comme un fil connectant deux parties d'un circuit
  - Il peut être de n'importe quel type (précisé lors de la déclaration)
  - Il se déclare dans l'architecture (avant le begin)
  - L'affectation d'un signal se fait après le begin

```
entity exemple is
   port (A,B,C: in std_logic;
        S: out std_logic);
end exemple;

architecture archi of exemple is
signal T: std_logic;
begin
   T <= A and B;
   S <= C or T;
end archi;</pre>
```

**C**3

## Signaux

#### REGLE FONDAMENTALE

 Un signal (ou un port de sortie) ne peut être affecté que par UNE SEULE instruction

```
architecture bad of exemple is
signal T: std_logic;
begin
   T <= '0';
   T <= '1';
end bad;</pre>
```

**C**3

## Types de Données

- Bit ('0', '1')
- Bit\_vector
- Boolean (false, true)
- Integer
- Natural (entiers ≥ 0)
- Positive (entiers > 0)
- Real
- Character
- String
- Time
- Tous ces types ne sont pas synthétisables (et servent juste à de la modélisation)

## Types IEEE

- Types de référence pour les bits/vecteurs de bits
- std\_logic
  - 9 états possibles
    - 'U': Non défini 🗷
    - 'X': Forçage Indéterminé (Conflit) 区
    - '0': Forçage au 0 logique ☑
    - '1': Forçage au 1 logique
    - 'Z': Haute impédance ☑
    - 'W': Niveau faible Indéterminé 区
    - 'H': Niveau 1 faible ☑
    - 'L': Niveau 0 faible ☑
    - '-': Peu importe ☑

☑ : Synthétisable

☑ : Non synthétisable

- std\_logic\_vector
  - Vecteur de std\_logic
  - signal A: std\_logic\_vector(3 downto 0); -- 4 bits

C3

#### Affectations

 Opérateur d'affectation pour les signaux et les ports



**C**3



28

- Affectation d'un entier
- Affectation d'un bit
- Affectation d'un vecteur de bits: C <= "001";</p>

: A <= 12;

: B <= '0';

## Opérateurs en VHDL

- Opérateurs logiques (bit à bit)
  - not
  - and
  - or
  - nand
  - nor
  - xor
  - xnor
  - (Concaténation)

```
A <= "110"; B <= "0011"; C <= B & A;
```

Opérateurs arithmétiques

```
-
```

- \* (synthétisable mais quelle architecture?)
- / (peu synthétisable)
- mod (pas synthétisable)
- exp (pas synthétisable)

**C3** 





## Langage Concurrent

 Par défaut, VHDL est un langage concurrent

 Cela signifie que toutes les instructions s'exécutent en parallèle

 Cela signifie aussi que l'ordre des instructions n'a pas d'importance

- 3 Niveaux sont proposés en VHDL
  - Flot de Données
    - Description bas-niveau
    - Écriture des équations logiques des sorties en fonction des entrées
  - Structurelle
    - Utilisation de composants pré-définis et référencés en bibliothèque
  - Comportementale
    - Description de plus haut niveau
    - Vision "algorithmique" ou fonctionnelle du système

**C**3

- Description flot de données
  - Rappel: Toutes les instructions s'exécutent EN MÊME TEMPS

```
library ieee;
use ieee.std logic 1164.all;
use ieee.std logic unsigned.all;
entity exemple is
  port (A,B,C: in std logic;
          S: out std logic);
end exemple;
architecture dataflow of exemple is
signal T: std_logic; -- SIGNAL INTERNE
begin
                                    L'ordre des
   S <= C or T; ←
                                    instructions
   T <= A and B;←
                                    est indifférent
end dataflow;
```

**C**3

#### Description structurelle

- 1 Description de modules VHDL
  - Après compilation, ces modules sont disponibles dans votre bibliothèque de travail

```
library ieee, work;
use ieee.std logic 1164.all;
use ieee.std logic unsigned.all;
entity AND Gate is
   port (A, B: in std logic;
         S: out std logic);
end AND Gate;
architecture dataflow of AND Gate
begin
   S \ll A and B;
end dataflow;
```

C3

Description structurelle

 2 – Réutilisation (Instanciation) des modules dans un autre composant

```
library ieee, work;
use ieee.std_logic_1164.all;
use ieee.std_logic_unsigned.all;
entity exemple is
   port (A,B,C: in std_logic;
        S: out std_logic);
end exemple;

architecture struct of exemple is
signal T: std_logic;
begin
   label0: entity work.OR_Gate(dataflow)
        port map (A=>C,B=>T,S=>S);
```

label1: entity work.AND Gate (dataflow)

port map (A=>A, B=>B, S=>T);

Utilisation d'un composant déjà inclus dans la bibliothèque

Connections des ports du composant instancié à ceux du module

Instanciation

\*
Instruction
concurrente

36

end struct;

- Instanciation:
  - 2 syntaxes possibles

```
entity module is
   port(H,Raz: in std_logic;
        E: in std_logic;
        S: out std_logic);
end module;
```

```
entity Top Level is
   port (Clock, Reset, E1, E2: in std logic;
         S1, S2: out std logic);
end Top Level;
architecture archi of Top Level is
begin
   -- Instanciation par Nom: l'ordre est indifférent
   Inst 1: entity work.module
   port map (Raz => Reset, S => S2, H => Clock, E => E2);
   -- Instanciation par Position: même ordre que l'entité module
   Inst 2: entity work.module
   port map (Clock, Reset, E1, S1);
end archi;
```

**C**3

- Description comportementale
  - Le système n'est plus vu comme une succession de portes
  - Vision plus fonctionnelle et algorithmique
    - Permet de décrire plus efficacement des systèmes complexes

```
library ieee;
use ieee.std logic 1164.all;
use ieee.std logic unsigned.all;
entity exemple is
   port (A, B, C: in std logic;
          S: out std logic);
end exemple;
architecture behavioral of exemple is
signal T: std logic;
begin
   S \ll T when C=0'
   else '1';
                                              L'ordre des
   T <= A when B='1'
                                              instructions
   else '0':
                                             est indifférent
end behavioral;
```

**C**3

## Syntaxe VHDL Concurrent

- Affectation conditionnelle
  - when else

```
s<="000" when E="00"

else "101" when E="01"

else "011";</pre>
```

**C3** 

with select

■ with E select

```
S <= "000" when "00",

"101" when "01",

"011" when others;
```

## Syntaxe VHDL Concurrent

- Duplication d'instructions
  - generate

```
entity AND Gate 4 is
    port ( a,b : in std logic vector(3 downto 0);
             s : out std logic vector(3 downto 0);
     end AND Gate 4;
                                                       S(0)
architecture arc of AND Gate 4 is
                                            A(1)
                                                       S(1)
begin
     LABEL 1: for i in 0 to 3 generate
                                                       S(2)
             s(i) \le a(i) and b(3-i);
     end generate;
end arc;
                                                       S(3)
```

**C**3

## Syntaxe VHDL Concurrent

- Duplication d'instanciations
  - generate

```
entity AND Gate 4 is
    port ( a,b : in std logic vector(3 downto 0);
            s : out std logic vector(3 downto 0);
    end AND Gate 4;
architecture arc of AND Gate 4 is
begin
    LABEL 1: for i in 0 to 3 generate
            LABEL 2: entity work.AND Gate 2
                     port map (a(i),b(3-i),s(i));
     end generate;
end arc;
```

C3

## Généricité

 VHDL permet de paramétrer les tailles des données d'un composant

 Intérêt: Réutilisation d'un même description pour plusieurs modules

Instruction generic

## Généricité


Exemple Valeur par défaut du paramètre entity Module is generic( N: natural:= 4); port (a,b : in std logic vector(N-1 downto 0); s : out std\_logic\_vector(N-1 downto 0)); end Module; architecture arc of Module is signal T: std\_logic\_vector(N-1 downto 0); begin T <= **not** A;  $S \le A$  and B; end arc;

## Généricité

Généricité et instanciation de composant

```
entity Module2 is
     port (a, b : in std logic vector(31 downto 0);
             s : out std logic vector(31 downto 0));
end Module2;
architecture arc of Module2 is
  signal T: std_logic_vector(31 downto 0);
  begin
      LABEL 2: entity work. Module
                generic map (32)
                port map (a,b,t);
                                           Changement
       S \leq not T:
                                           de la valeur
end arc;
                                          du paramètre N
```

C3



# VHDL Concurrent/Séquentiel

- Les descriptions vues précédemment sont appelées VHDL concurrent
  - Pour certains composants, cette méthode de description n'est pas toujours la plus agréable à utiliser
- VHDL offre la possibilité de décrire séquentiellement le comportement d'un système
- Cela se fait dans une structure de programmation appelée process
- On parle alors de VHDL séquentiel

**C**3

# VHDL Concurrent/Séquentiel





- Électronique combinatoire/séquentielle
- Description VHDL concurrente/séquentielle

- On peut décrire du combinatoire en VHDL séquentiel
- On peut décrire du séquentiel en VHDL concurrent
- Et vice et versa...
- Seuls la syntaxe et le style de programmation changent...

**C**3

#### **Process**

- Un process peut être vu comme une fonction dont l'exécution dépend de la variation de signaux/ports à observer
- Il permet une description plus algorithmique du système.
- Les signaux/ports que l'on souhaite observer font partie de la liste de sensibilité du process
  - > Si un des signaux de la liste est modifié, le process s'exécute
  - > Si aucun des signaux n'est modifié, le process reste en veille

**C**3

## Liste de sensibilité

#### REGLE FONDAMENTALE

 Lorsque l'on décrit dans un process un système électronique combinatoire, la liste de sensibilité doit comporter TOUTES LES ENTREES du système

C3

 Lorsque l'on décrit dans un process un système électronique séquentiel, la liste de sensibilité doit comporter L'HORLOGE et les ENTREES DE FORCAGE ASYNCHRONE (reset, preset)

## Exemple

Description d'une bascule D

```
library ieee;
use ieee.std logic 1164.all;
                                                 Liste de sensibilité
use ieee.std logic unsigned.all;
                                               process s'exécute si
                                               évolution des signaux
entity Bascule is
port(h, raz, d: in std logic;
     q: out std logic);
end Bascule;
                                               Test du front montant
architecture archi of Bascule is
begin
   process (h, raz)
   begin
        if raz='0' then g = '0';
        elsif rising edge(h) then q<=d;</pre>
        end if;
   end process;
end archi;
```

C3

### Process VHDL et Simulation

- Comment un simulateur VHDL effectue-t-il:
  - La gestion d'un process
  - La gestion de plusieurs process
    - (avec des listes de sensibilité différentes)
  - La gestion d'un process associé à des instructions concurrentes

### Simulateur VHDL

- Simulation à Evénements Discrets
  - A chaque pas de simulation
    - Evolution des valeurs des entrées
    - Ces changements activent les listes de sensibilités de certains process
    - Exécution de ces process
      - Dans un ordre quelconque
      - SANS MISE A JOUR IMMEDIATE DES SORTIES
    - Une fois ceci fait, mise à jour simultanée des sorties des process puis des instructions concurrentes
      - Ce mécanisme garantit la bonne synchronisation du modèle
    - Détermination du prochain pas de simulation

**C3** 

#### **Process et Simulation**

- Pour garantir l'aspect concurrent d'un programme
   VHDL et la bonne synchronisation des instructions,
   un process possède les caractéristiques suivantes:
  - Pour le simulateur, le temps d'exécution d'un process est nul (sauf si un délai y est explicitement indiqué)
  - Lors de la simulation du programme, l'ensemble des instructions d'un process est vu comme une grosse instruction concurrente
  - Pour le simulateur, deux process ayant la même liste de sensibilité seront activés aux mêmes pas de simulation.
  - Un signal/port dont la valeur est modifiée au cours d'un process n'est remis à jour QU'APRES L'EXECUTION DE TOUS LES PROCESS

**C**3

## Exemple 1


```
entity Exemple is
                                       PROCESS 1
                                                            PROCESS 2
port(clk,e: in std logic;
     s: out std logic);
end Exemple;
architecture archi of Exemple is
signal t: std logic;
                                Clk
begin
   process (clk)
   begin
         if rising edge(clk) then t<=e; end if;</pre>
   end process;
   process (clk)
   begin
         if rising edge(clk) then s<=t; end if;</pre>
   end process;
end archi;
                           Quel est le comportement des signaux T et S?
```

C3

54

L3 EEA - LU3EE100

## Exemple 1 - Simulation



```
C3
```




## Exemple 2

Affectation et mise à jour d'un signal dans un process

```
entity Compteur is
                                               Le test se fait avec
port(h, raz: in std logic;
                                           la valeur qu'a le signal cpt
     sortie: out std logic);
                                              en DEBUT de process
end Compteur;
architecture archi of Compteur is
signal cpt: std logic vector(2 downto 0);
begin
   process(h, raz)
   begin
         if raz='0' then cpt<="000"; sortie<='0';</pre>
         elsif rising edge(h) then
             cpt<=cpt+1; /
             if cpt="110" then sortie<='1'; else sortie<='0'; end if;</pre>
         end if:
   end process;
                                             Mise à jour de cpt
end archi;
                                                et de sortie
```

**C3** 


## Exemple 2 - Simulation



**C**3



# Architecture synthétisée



**C**3

"Sortie" est synthétisé sous la forme d'une bascule D

> REGLE:

Tous les signaux/ports qui sont affectés dans un process sensible sur une horloge seront synthétisés sous forme de bascules/registres

#### **Variables**

Le process dispose d'une structure de données propre: la variable

 Une variable sert à stocker un résultat intermédiaire dans le déroulement d'un algo

 Contrairement au signal, une variable dont la valeur est modifiée est IMMEDIATEMENT remise à jour

 Il n'est pas possible d'utiliser une variable en dehors d'un process

**C**3

#### **Variables**

#### ■ Exemple – Détection de parité

```
entity Parity is
                                                       Déclaration
port(x: in std logic vector(7 downto 0);
                                                      d'une variable
     p: out std logic);
end Parity;
architecture archi of Parity is
begin
                                              Opérateur d'affectation
   process(x)
                                                   différent de
        variable tmp : std logic := '0';
                                                 celui des signaux
   begin
        for i in 0 to 7 loop -- boucle for
          if x(i) = 1' then tmp := not tmp; end if;
        end loop; -- fin de la boucle for
                                                    Affectation
        p <= tmp; ←
                                                de la variable dans
   end process;
                                              un signal/port de sortie
end archi;
```

C3

60

L3 EEA - LU3EE100

## Signal vs Variable

#### SIGNAL

- Portée:
  - Tout le programme
- Opérateur d'affectation:

<=

Mise à jour

VHDL concurrent:

Immédiate

VHDL séquentiel (process)

- A la sortie du process
- INTERET

Description de matériel

### VARIABLE

- Portée:
  - Locale à un process
- Opérateur d'affectation:

:=

Mise à jour

VHDL concurrent:

Impossible

VHDL séquentiel (process)

- Immédiate
- INTERET

Stockage d'un résultat intermédiaire d'un algo

61

LE357 - 2009-2010

# Instructions VHDL Séquentiel

```
■ IF - THEN - ELSIF - ELSE
library ieee;
use ieee.std logic 1164.all;
use ieee.std logic unsigned.all;
entity Mux41 is
port(a,b,c,d: in std logic vector(7 downto 0);
     com: in std logic vector(1 downto 0)
     s: std logic vector(7 downto 0));
end Mux41;
architecture archi of Mux41 is
begin
   process(a,b,c,d,com)
   begin
        if com="00" then s<=a;
        elsif com="01" then s<=b;</pre>
        elsif com="10" then s<=c;
        else s<=d;</pre>
        end if:
   end process;
end archi;
```

CJ

# Instructions VHDL Séquentiel

#### CASE – WHEN

```
entity Mux41 is
port(a,b,c,d: in std logic vector(7 downto 0);
     com: in std logic vector(1 downto 0)
     s: out std logic vector(7 downto 0));
end Mux41;
architecture archi of Mux41 is
begin
   process(a,b,c,d,com)
   begin
        case (com) is
            when "00" => s<=a;
            when "01" => s<=b;
            when "10" => s<=c;
            when "11" => s<=d;
                                           Rappel: un std_logic
            when others => NULL;
                                           a 9 états possibles
        end case;
   end process;
                                          Erreur à la compilation
end archi;
                                          si instruction absente
```

# Instructions VHDL Séquentiel

```
■ FOR - LOOP
library ieee;
use ieee.std logic 1164.all;
use ieee.std logic unsigned.all;
entity exemple for is
port(a,b: in std logic vector(7 downto 0);
     s: std logic vector(7 downto 0));
end exemple for;
architecture archi of exemple for is
begin
  process (a,b)
  begin
       for i in 0 to 7 loop
              s(i) \le a(7-i) and b(i); B
       end loop;
  end process;
end archi;
```



# Bien programmer en VHDL

Ce qu'on doit faire, ce qu'il vaut mieux éviter...

- VHDL n'est pas case sensitive
  - Pas de différence entre minuscules et majuscules

- Commentaires
  - \_\_\_
  - La suite de la ligne est en commentaires

Initialisation de signaux

```
signal toto: std_logic_vector(2 downto 0) := "101";
```

- La valeur initiale sera prise en compte en simulation
- Ce ne sera pas forcément le cas en synthèse
  - > Cela dépend de l'outil
- Une initialisation "propre" d'une information dans signal ou d'un port de sortie se fait à l'aide d'un reset asynchrone...



- Types Entiers (integer, natural, positive)
  - Les données de ces types sont par défaut sur 32 bits et synthétisés ainsi, même si la pleine résolution n'est pas utile
  - Pour ajuster la taille des entiers aux besoins du modèle, il faut utiliser l'expression range

```
signal TOTO: integer range 0 to 7;
     -- Synthétisera un entier sur 3 bits
TOTO <= 3; -- OK
TOTO <= 12; --KO (out of range...)</pre>
```

**C**3

Pour affecter une partie d'un vecteur à 1 et le reste à 0

```
signal titi: std_logic_vector(7 downto 0);

titi <= (7 => '1', 4 downto 2 => '1', others => '0');

-- idem titi <= "10011100";

Pas de parenthèses</pre>
```

C3

Pour tout affecter à 0

dans ce cas...

```
titi <= (others => '0'); -- idem titi <= "00000000"
```

Affectation de valeurs hexadécimales

```
signal toto: std_logic_vector(7 downto 0);
toto <= x"3C"; -- idem toto <="00111100";</pre>
```

 Attention: Le nombre de bits du vecteur doit être un multiples de 4

```
signal titi: std_logic_vector(6 downto 0);
titi <= x"3C"; -- Erreur à la Compilation!</pre>
```

Instanciation

```
entity module is
   port(H,Raz: in std_logic;
        E: in std_logic;
        S: out std_logic);
end module;
```

```
end module;
entity Top Level is
   port (Clock, Reset, E1, E2: in std logic;
         S1, S2: out std logic);
end Top Level;
architecture archi of Top Level is
begin
   Inst 1: entity work.module
   port map (Clock, Reset, E1, S1);
   -- Instanciation par Position: même ordre que l'entité module
   Inst 2: entity work.module
           port map(Raz => Reset, S => S2, H => Clock, E => E2);
   -- Instanciation par Nom: l'ordre est indifférent
```

71

end archi;

#### Attributs

- Permet d'extraire une caractéristique d'un signal/port/variable X
- Syntaxe: X'attribut

#### Exemples:

```
signal toto: std_logic_vector(N-1 downto 0);
-- N paramètre générique
if (toto'high = 4) then ...
-- Est-ce que ma donnée est sur 5 bits?
```

signal horloge: std\_logic;

```
if (horloge'event and horloge = '1') then ...
-- Idem rising_edge(horloge)
```

**C**3

# Règles générales

- Un signal/port doit être décrit à UN et UN SEUL endroit
- Pourquoi?
  - Car TOUTES les instructions concurrentes s'exécutent en parallèle

```
architecture archi of erreur1 is
begin

a <= '0';
s1 <= c and d;
s2 <= e or c;

a <= b when e= '1'
else c;

end archi;</pre>
```

**C**3

# Règles générales

- Un signal/port est décrit soit dans UN SEUL process, soit dans UNE SEULE instruction concurrente
- Pourquoi?
  - Car un process est vu comme une grosse instruction concurrente


```
architecture archi of erreurl is
begin

a <= '0';

process(b,c,e)
begin
    if e= '1' then a<=b;
    else a<=c;
    end if;
end process;
end archi;</pre>
```

**C**3

 De même, une instanciation de module est considéré comme une instruction concurrente




S2 est affecté à la sortie du bloc MODULE, on ne peut donc pas ajouter d'instruction type S2<= toto;

**C3** 

- Listes de sensibilité d'un process:
  - Système combinatoire:
     TOUTES LES ENTREES du combinatoire
  - Système séquentiel:HORLOGE + Reset/Preset ASYNCHRONE
- Ne pas tenter de mélanges:
  - Ne pas décrire un système séquentiel dans un process qui n'est pas déclenché par l'horloge
  - Et inversement...


**C**3

- Dans un process
  - Une variable est mise à jour IMMEDIATEMENT
  - Un signal est mis à jour EN SORTIE DU PROCESS



**C3** 

- Ports d'entrée/sortie
  - On ne peut pas affecter (écrire) une valeur sur un port d'entrée
  - On ne peut pas lire la valeur d'un port de sortie pour l'affecter à un signal ou un autre port



**C**3


- Les valeurs des sorties doivent être définies pour toutes les combinaisons possibles des entrées
  - Démonstration à l'aide d'un contre exemple
    - Multiplexeur 3 → 1

```
entity mux_3_1 is
port(a,b,c: in std_logic;
    com: in std_logic_vector(1 downto 0);
    s: out std_logic);
end mux_3_1;
architecture archi of mux_3_1 is
begin
    s <= a when com="00"
    else b when com="01"
    else c when com="10";
end archi:</pre>


Que se passe-t-il
    si com="11"?
```

**C**3

Architecture souhaitée



Architecture générée



- **C**3
- 80

- Si Com=11 → l'état de S est maintenu
- Insertion d'un latch D
- ➤ Latch difficilement synthétisables sur FPGA
- > Dysfonctionnement après implémentation

- Les valeurs des sorties doivent être définies pour toutes les combinaisons possibles des entrées
  - Description correcte (synthétisable)
    - Multiplexeur 3 → 1

```
entity mux_3_1 is
port(a,b,c: in std_logic;
        com: in std_logic_vector(1 downto 0);
        s: out std_logic);
end mux_3_1;
architecture archi of mux_3_1 is
begin
        s <= a when com="00"
        else b when com="01"
        else c;
end archi;</pre>
```

**C**3

Si le combinatoire est décrit dans un process,
 mettre toutes les entrées dans la liste de sensibilité

#### Pourquoi?

- La sortie d'un combinatoire est directement dépendante de la combinaison des entrées
- Tout changement sur une entrée entraîne potentiellement un changement de la sortie
- Si une entrée ne figure pas dans la liste du process, un changement de valeur sur cette entrée ne sera pas immédiatement détecté
  - -> Mauvaise description du comportement de la sortie








## Bien décrire du séquentiel

 La liste de sensibilité comprend UNIQUEMENT l'horloge et le reset asynchrone

#### Pourquoi?

- La sortie d'une bascule est modifiée lors d'un front d'horloge ou lors de l'activation du reset asynchrone
- Le changement de son entrée n'entraîne pas IMMEDIATEMENT un changement en sortie



**C3** 

## Bien décrire du séquentiel

- A quoi sert le reset ASYNCHRONE?
  - A donner des valeurs aux bascules lors de l'init. du système
  - Il n'y a ainsi pas de valeurs non définies au démarrage

- Quelle différence avec un reset SYNCHRONE?
  - Reset asynchrone:
    - Initialisation globale du système
    - Gérée par l'utilisateur (ON/OFF)
  - Reset synchrone:
    - Réinitialisation d'une partie du système
    - Auto-gérée par le système








## Bien décrire du séquentiel

- Quel est l'ordre des tests dans une description d'un système séquentiel?
  - Le RAZ asynchrone est prioritaire sur l'horloge
    - On le teste donc en premier
  - On teste ensuite le front d'horloge
  - On teste enfin les autres entrées de commande

```
begin
    if raz='0' then registre<="0000";
    elsif rising_edge(h) then
        if load='1' then registre<=entree;
        else registre<="0011"; end if;
    end if;
end process;</pre>
```

C3

- Exemple d'un système mêlant combinatoire et séquentiel
  - Additionneur avec registres d'entrée et de sortie



**C**3

Ce qu'il ne faut pas faire

Décrire l'additionneur dans le même process que les registres (avec l'horloge comme liste de sensibilité)

**C3** 

87

d'affectation sont des bascules/registres

Ce qu'il ne faut pas faire


Décrire l'additionneur dans le même process que les registres (avec l'horloge comme liste de sensibilité)

C3

88

L'additionneur n'est plus combinatoire

Architecture réalisée par le programme précédent



**C**3

- Ce qu'il faut faire
  - Décrire les parties séquentielles et les parties combinatoires dans des zones différentes du programme
  - Par exemple, décrire le séquentiel dans un ou plusieurs process sensible(s) sur l'horloge et le reset
  - Et décrire le combinatoire à l'aide d'instructions concurrentes

**C3** 

Programme correct

```
-- ZONE COMBINATOIRE DU PROGRAMME
Adder <= '0' & RegA + RegB;
-- ZONE SEQUENTIELLE DU PROGRAMME (PROCESS)
process(h, raz)
  begin
       if raz='0' then RegA<="0000"; RegB<="0000";</pre>
                              RegS<="00000";
       elsif rising edge(h) then
               RegA <= EntreeA;</pre>
               ReqB <= EntreeB;</pre>
               RegS <= Adder;</pre>
       end if;
  end process;
```

C3

Autre exemple: compteur

Mauvaise description d'un compteur (module séquentiel) dans un process combinatoire

```
begin
   if start='1' then compteur<=entree;
   else compteur <= compteur + 1;
   end if;
end process;

Boucle combinatoire: l'incrémentation
   se répète indéfiniment tant que
   le test sur start ='0' est validé.
   NON SYNTHETISABLE!!</pre>
```

**C**3

Programme correct

Description dans un process sensible sur l'horloge

```
Permet d'affecter tous les
    bits d'un vecteur à '0'

begin


if reset='0' then compteur<=(others =>'0');
elsif rising_edge(h) then
    if start='1' then compteur <= entree;
else compteur <= compteur+1;
end if;
end if;
end process;

Grâce au test du front d'horloge,</pre>
```

l'instruction d'incrémentation ne s'exécutera

qu'UNE SEULE FOIS par cycle d'horloge

C3



## Testbenchs, Simulation

Comment simuler efficacement son code VHDL

#### Outils de Simulation VHDL

- La plupart des outils permettent de simuler les modèles VHDL de deux façons différentes
  - Simulation "manuelle"
    - Réalisation de la simulation étape par étape
    - Chaque étape est paramétrée "à la main" par l'utilisateur
  - Simulation par testbench
    - Ecriture préalable du déroulement de la simulation dans un fichier appelé banc de test
    - L'outil exécute d'un bloc le scénario du banc de test
- Le simulateur EDA Playground ne fonctionne qu'avec des testbenchs

**C3** 

#### Simulation Manuelle

- Chaque étape de simulation s'effectue en trois temps
  - On donne une valeur aux entrées
  - On fait avancer le temps de simulation
  - On observe les valeurs des sorties

**C3** 

Puis on itère ces trois étapes pour toutes les combinaisons possibles...

### Inconvénients Simu. Manuelle

- Si le système possède 32 bits en entrée
  - 4 294 967 296 combinaisons possibles à tester
  - Sans compter les séquences de valeurs...

#### Conséquence

- La simulation manuelle va être TRES longue
- La probabilité que vous fassiez une erreur est TRES importante
- Conséquence d'une erreur
  - Il faut tout recommencer depuis le début...

**C3** 

## Comment échapper à cela?

#### Faire un Testbench

 Fichier dans lequel on va écrire les valeurs successives que vont prendre les entrées au cours de la simulation

**C3** 

#### Testbench en VHDL

- C'est un programme VHDL
- On y instancie le module à simuler
- Puis on y décrit le comportement des entrées

### Exemple 1 - Testbench

```
library ieee;
                                             MODULE
use ieee.std logic 1164.all;
use ieee.std logic unsigned.all;
entity TestBench is
end TestBench;
                                           Com
architecture simu of TestBench is
signal entree, sortie: std_logic vector(3 downto 0);
signal commande: std logic;
begin
  L0: entity work. Module
  port map (E=>entree, C=>commande, S=>sortie);
   entree<="0100", "1001" after 100 ns, "1111" after 200 ns;
   commande <= '0', '1' after 40 ns, '0' after 120 ns;
end simu;
```

**C**3

### Exemple 1 - Testbench


```
library ieee;
use ieee.std logic 1164.all;
use ieee.std logic unsigned.all;
                                    Instanciation du composant
entity TestBench is
end TestBench;
                                          VHDL à simuler
architecture simu of TestBench is
signal entree, sortie: std_logic_vector(3 downto 0);
signal commande: std logic;
begin
                                              Evolution des
   L0: entity work. Module
  port map (E=>entree, C=>commande, S=>sort signaux d'entrée
   entree<="0100", "1001" after 100 ns, "1111" after 200 ns;
   commande <= '0', '1' after 40 ns, '0' after 120 ns;
end simu;
```

### Exemple 1 - Testbench

Chronogramme des entrées du Testbench

**C**3





## Exemple 2 - Testbench

```
library ieee;
                                              MODULE
use ieee.std logic 1164.all;
use ieee.std logic unsigned.all;
entity TestBench is
end TestBench;
architecture simu of TestBench is
signal entree, sortie: std logic vector(3 downto 0);
signal clk: std logic:='0';
                                                  Valeur initiale
begin
                                                   donnée au
                                                 signal d'horloge
   L0: entity work. Module
  port map (E=>entree, H=>clk, S=>sortie);
   entree<="0100", "1001" after 100 ns, "1111" after 200 ns;
   clk<= not clk after 100 ns;
                                                Evolution
end simu;
                                               de l'horloge
```


C3

### Exemple 2 - Testbench

Chronogramme des entrées du Testbench

**C3** 





#### Utilisation du Testbench

- Compilation du module à tester
- Compilation du module testbench
- Lancement du simulateur pour simuler le testbench
- On fait avancer le temps
  - Les entrées évoluent automatiquement
  - On observe le comportement des sorties
- Si on s'aperçoit qu'on a fait une erreur
  - On modifie les instructions incorrectes
  - On recompile et on relance la simulation
- Gain de temps considérable

**C**3



## Multiplexeur

Exemple de multiplexeur 4 → 1

```
entity mux 4 1 is
port(a,b,c,d: in std_logic;
     com:in std logic vector(1 downto 0);
     s: out std logic);
end mux 4 1;
architecture archi of mux 4 1 is
begin
                                           Com(1:0)
  s <= a when com="00"
  else b when com="01"
  else c when com="10"
  else d;
end archi;
```

C3

## Comparateur

Exemple avec entrées sur 4 bits

```
entity comparateur is
port(a,b: in std_logic_vector(3 downto 0);
     egal, sup, inf: out std logic);
end comparateur;
architecture archi of comparateur is
begin
                                           Comparateur
       egal <= '1' when a=b else '0';
       sup <= '1' when a>b else '0';
       inf <= '1' when a <b else '0';
end archi;
                                           sup
                                               egal
```


112

Voir les détails de l'architecture dans le poly précédent...

### Incrémenteur

```
entity incrementeur is
port(e: in std logic vector(3 downto 0);
     s: out std logic vector(3 downto 0));
end incrementeur;
architecture archi of incrementeur is
begin
      s <= e + 1;
end archi;
                                   Expression utilisable
                                    grâce au package
                                   std_logic_unsigned
```

#### Additionneur



L3 EEA - LU3EE100

#### Additionneur

Avec ou sans retenue sortante

```
entity adder is
port (
  a,b: in std logic vector(3 downto 0);
  cin: in std logic;
  s1: out std logic vector(3 downto 0); -- sans retenue
  s2: out std logic vector(4 downto 0)); -- avec retenue
end adder;
architecture archi of adder is
begin
  s1 <= a+b+cin; -- Addition sans retenue sortante
  s2 <= '0'&a+b+cin; -- Addition avec retenue sortante
end archi;
```

## Registre simple

```
entity Registre is
port( h, raz: in std logic;
       entree: in std logic vector (3 downto 0);
       sortie: out std logic vector(3 downto 0));
end Registre;
architecture archi of Registre is
begin
  process(h, raz)
  begin
       if raz='0' then
              sortie<=(others => '0');
       elsif rising edge(h) then
              sortie <= entree;</pre>
       end if;
  end process;
end archi;
```

C3

## Registre multifonctions (1/2)

```
entity Registre is
       h, raz, serial: in std logic;
port(
        com: in std logic vector(1 downto 0);
        E: in std logic vector(7 downto 0);
        S: out std logic vector(7 downto 0));
end Registre;
                                                     Fonction
                                                              Com
Serial
                                       Reg
                                                     Mémo
                                                                0
                                                     Chgt //
                                                     Decal \rightarrow
                                                                3
                                                    RAZ Sync

    Schéma Bloc
```

**C**3

## Registre multifonctions (2/2)

```
architecture archi of Registre is
signal reg: std logic vector(7 downto 0);
begin
 S <= reg; -- CONNECTION DU REGISTRE AU PORT DE SORTIE
process (h, raz)
begin
  if raz='0' then reg<=(others => '0'); -- RAZ ASYNCHRONE
  elsif rising edge(h) then
   case (com) is
     when "01" => reg<=E; -- Chgt //</pre>
    when "10" => reg<=serial & reg(7 downto 1); -- DECALAGE</pre>
     when "11" => reg<=(others => '0'); -- RAZ SYNCHRONE
   end case;
  end if:
 end process;
end archi;
```


C3

## Compteur simple

```
entity Compteur is
port( h, raz: in std logic;
        cpt: out std logic vector(3 downto 0));
end Compteur;
architecture archi of Compteur is
signal reg: std logic vector(3 downto 0);
begin
   cpt <=reg;</pre>
   process (h, raz)
  begin
      if raz='0' then
         req <= (others => '0');
      elsif rising_edge(h) then
         req \leq req+1;
      end if:
   end process;
end archi;
```

C3

## Compteur multifonctions (1/2)



C3

## Compteur multifonctions (2/2)

```
architecture archi of Compteur is
signal reg: std logic vector(7 downto 0);
begin
  cpt <=req;
  process (h, raz)
  begin
     if raz='0' then reg <= (others =>'0');
     elsif rising edge(h) then
       case(com) is
         when "01" => reg<=reg+1; -- Incrementation</pre>
         when "10" => reg<=reg-1; -- Decrementation</pre>
         end case;
     end if:
  end process;
end archi;
```

C3